correction in listing — see Abstract

Mathematics Notes
Note II
10 November 1966

FORPLEX

A Program to Calculate Inverse Fourier Transforms

Frank J. Sulkowski
The Dikewood Corporation

ABSTRACT

This paper describes a computer program that calculates the Inverse Fourier Transform

$$F(t) = \frac{1}{2\pi}\int_{-\infty}^{\infty} G(\omega)\, e^{j\omega t}\, d\omega \quad ,$$

where the input, $G(\omega)$, is complex and $\omega = 2\pi f$. An input parameter determines the accuracy to which the computation is carried.

The system of subroutines was programmed for the CDC Chippewa Fortran System for use on the CDC-6600 at the Air Force Weapons Laboratory at Kirtland Air Force Base. The work was performed under Air Force Contract Number AF 29(601)-7263.

FORPLEX

A PROGRAM TO CALCULATE INVERSE FOURIER TRANSFORMS

INTRODUCTION

The calculation of F(t) by FORPLEX is iterative in that two passes are made through the subroutines making up the program. Numerical integrating techniques devised by Stanford Research Institute (Appendix A) and initial programs were furnished by Mr. G. Carpenter of SRI. These techniques were modified so that a complex function, or data, could be used for input, also, so that angular frequency ($2\pi f$) could be used in the calling program.

The program input consists of the complex function $G(\omega)$, as either data points or an analytic function to be transformed; a criterion, sigma to establish the accuracy of the transform; arrays containing time steps and the time interval end points over which the steps are used in the integration; and the number of these pairs. Other communication to the subroutines is through the use of labeled common areas. All printed output, comparisons, parameter generation, etc., are the responsibility of the calling program.

The use of these subroutines is general purpose in nature. Effective use of the delta time (time step) and ending time point arrays enables one to examine in detail the behavior of a complex function in a specific region and to minimize oscillatory behavior of the integration at both end points for the times of interest.

The input parameter, sigma, determines the upper and lower bounds in the frequency domain; and also a frequency step, delta omega, for each decade range, in subranges of a 1-2-5 nature to maintain a logarithmic relationship. Finally, an error or difference calculation is returned to the calling program which is of the form $(\Sigma \Delta V_i^2 \Delta t_i \,/\, \Sigma V_i^2 \Delta t_i)^{\frac{1}{2}}$ where V_i is the calculated value of F(t) over the i^{th} time interval, and ΔV_i is the difference of this value and the value of F(t) calculated previously over that interval. In the second iteration, or pass, the range in the frequency domain is extended in each direction (upper, and lower decades) and the input criterion, sigma, is replaced by $\sigma/2$, to obtain a finer integrating grid in each decade.

OPERATION

General

The sample calling program listed in Appendix B calculates F(t) for $G(\omega)$, where $G(\omega) = \frac{1}{1+j\omega}$, and F(t) is the known function (e^{-t}). The input criterion, sigma, is 0.1, therefore the subroutines return an approximation of F(t) for $\sigma/2$. Delta time (Δt) is from 0.02 to 2.5 time units and there is one pair of Δt and ending time. Note that these are communicated to the subroutines through the arrays T and DT in labeled common, CBFOR.

The function or data are calculated for $G(\omega)$ in the complex array (RCVC). The array OM(I) contains the omega values; T(I) contains the appropriate time value that is generated by both Δt 's and ending time values; VN(I) holds the first pass results of F(t); and V(I) holds the final pass results of F(t) after extending the range of omega.

For each run, the size of the communicating arrays must be determined. The T, VN and V arrays are determined by the approximation $1 + \Sigma_i \left(\frac{t_i \text{ ending value}}{\Delta t_i \text{ in that range}} \right)$. The OM, RCVC arrays are determined by the approximation, JFREQ = $(3.5/\sigma)$ times the number of decades used. The lowest decade is 10^L, where L is the integral part of $\log_{10}(\sigma/\text{time range}) - 1$. That is, the lowest decade is approximately (1/full time range) minus one decade. The highest decade is 10^H, where H is similarly the integral part of $\log_{10}(1/(\sigma \times \text{smallest } \Delta t)) + 1$ or simply, the highest decade is (1 / smallest Δt) + one decade.

The arrays DT(I), TIM(I) are for the Δt 's and the ending times for which they are used. NT is the number of pairs of these values used.

Methods

The first call to the subroutines is through the subroutine FIPI (Fourier Integral Parameter Input) at which time the parameters sigma, NT, and the arrays for delta time and ending time are transmitted to the using subroutines. The parameter JFREQ is a return parameter which is the number of values to be calculated for the function $G(\omega)$, as indicated by the array RCVC.

The complex function RCVC or $G(\omega)$ must be calculated twice by the calling program, therefore the switch (JGO) is set to one initially. As an example for $G(\omega) = \frac{1}{1+j\omega}$, RCVC(I) = (1., 0.)/CMPLX(1., OM(I)). This is done for each omega OM(I), which are supplied by the subroutine FIPI with the input parameter sigma.

With the calculation of RCVC, the subroutine FIOFI (Fourier Integral Omega Function Input) is called and the first calculation of F(t) is made; also the decade range is extended and the criterion divided by two. Setting the switch (JGO) to two, the complex function $G(\omega)$ is again calculated as shown in the sample listing.

FICO (Fourier Integral Calculated Output) is then called for a final calculation of F(t). FICO also returns the RMS or difference comparison of both F(t 's) and the index for the number of quantities calculated.

The sample calling program then makes use of all calculations to print differences and a comparison with a known function, in this case $F(t) = e^{-t}$. The sample calling program for calculating the inverse transform of $1/(1+j\omega)$ is an illustration only. The program can easily be modified so that the calling subroutines are in a major iteration loop. Also, the size of the communicating arrays in the labeled common statements can be easily changed. If the labeled common statements are modified, they must also be modified in the subroutines dependent on them.

SUBROUTINES

Six subroutines make up this program. As noted, communication is made with these subroutines through labeled common areas and through input parameters from the calling program.

The subroutine FIPI determines delta omega and the range of omega with the input parameters delta time, its ending time value, and sigma. FIPI uses two minor subroutines WRSET (Omega Range Set) and WSET (Omega Set) to set up the elements in the arrays used by the main subroutine CALCLT.

FIOFI calls the main subroutine CALCLT (Calculate) which calculates F(t). FIOFI also redefines delta omega and omega through WRSET and WSET and their number JFREQ, and also halves the input criterion.

FICO again calls CALCLT and then computes the comparison of F(t) with the previously calculated F(t). The number of values for F(t) is also returned to the calling program by the parameter (JTIM). The program running time is approximately 80 points per second where the number of points processed is given by JTIM+JFREQ(1)+JFREQ(2). JFREQ(1) is used when the switch JGO is one and JFREQ(2) is used when JGO is two.

APPENDIX A
SPECTRUM CALCULATIONS

For numerical integration of the Fourier integral

$$F(f) = \int_{-\infty}^{\infty} F(t)\, e^{-j2\pi ft}\, dt \quad ,$$

one simply divides the time base into a number of intervals, measures a representative amplitude at the center of the interval, multiplies by the width of the interval and by a phase factor, and then sums over all intervals. In the limit, as the intervals are made smaller, the summation provides a reasonably accurate spectrum of the time function. This was the calculation program on hand at SRI when the EMP project began, and at that time it seemed adequate. The program was used for two sample waveforms to determine the limitations. It was found that accurate calculations of the high-frequency content required small interval widths on the time base over the entire waveform.

The reason that small sample spaces are required for numerical integration, even in the low-frequency part of the waveform, is that a representative amplitude and phase factor are chosen for each interval, and these no longer remain representative for calculations at high frequencies. Calculations at high frequencies effectively cause a large change in phase factor from one side of the interval to another, and the value at the center is thus no longer representative. This problem can be overcome if one approximates the amplitude function over each interval by some simple mathematical function, performs the Fourier integral calculation for each interval, and then sums the results. The preceding amounts to assuming that

$$F(t) = f_1(t) + f_2(t) + f_3(t) + \cdots + f_n(t) \quad ,$$

where $F(t)$ is the expression for the entire waveform and $f_1(t)$, $f_2(t)$, etc. are approximate functions valid in intervals 1, 2, etc.

The simplest approximation that can be used is to assume that each $f_i(t)$ is a constant equal to the amplitude of the waveform measured at the center of the interval. This gives rise to the "stepped" approximation shown in the following sketch:



D-5585-26

STEPPED MODEL

The next degree of improvement is to use a constant slope between amplitude values measured at the edges of each interval. This gives rise to the "straight-line" approximation shown in the following sketch:



D-5585-27

STRAIGHT-LINE MODEL

A marked improvement in the fidelity of the model waveform is obtained by going from the stepped to the straight-line approximation. Further improvement can be obtained by using second and higher-order approximations, but the problem of generating and integrating these functions becomes increasingly more difficult.

Uniform intervals were shown in the two previous sketches, but in fact each interval could have been made a different size. The derivations that follow show that there are some advantages in using a compromise between all uniform intervals and all unequal intervals. The compromise demonstrated here is the use of several sets of uniform intervals, where the interval of one set is a different length from that of another set. The two fundamental features incorporated in the spectrum calculations that follow are that a straight-line model of the original waveform is used, and that several sets of uniform intervals of different size are used.

The nomenclature appropriate to the waveform model is as follows: A number of amplitude-time pairs received from the scaling operation are denoted A_i, t_i. The fundamental unit of time considered in the discussion is the smallest sampling interval, Δt_1, which will almost always be used for the earliest part of the waveform. (The smallest interval will determine the maximum frequency at which the waveform model will follow the spectrum of the actual waveform.) We now define $t_i = i\Delta t_1$, true for all times in the model. From this definition it is evident that when we come to a change in the scaling interval, the subscripts of t are no longer consecutive integers. Consider, for example, that at t_{10}, the scaling interval changes from Δt_1 to Δt_2 were, say, $\Delta t_2 = 2\Delta t_1$; then the amplitude-time pairs would have the following notation: A_8, t_8; A_9, t_9; A_{10}, t_{10}; A_{12}, t_{12}; A_{14}, t_{14}; etc.

A fact that will be used often in the following discussions is that the straight-line model of a waveform can be shown to be made up of a number of isoceles triangles when uniform intervals are used. If the individual triangles in the following sketches are added, their sum yields the straight-line model of a waveform in the interval of time shown.



D-5585-25

TRIANGLE SUMMATION

The Fourier integral pair to be used here is defined as follows:

$$F(f) = \int_{-\infty}^{\infty} F(t)\, e^{-j2\pi ft}\, dt$$

$$F(t) = \int_{-\infty}^{\infty} F(f)\, e^{j2\pi ft}\, dt \quad .$$

Using this definition, the spectrum of an isoceles triangle of height A_i, centered at $t_i = i\Delta t_1$, and of width $2\Delta t_1$ is

$$F(f) = \int_{t_i-\Delta t_1}^{t_i} A_i \left[\frac{t-t_i+\Delta t_1}{\Delta t_1}\right] e^{-j2\pi ft}\, dt + \int_{t_i}^{t_i+\Delta t_1} A_i \left[\frac{t_i+\Delta t_1-t}{\Delta t_1}\right] e^{-j2\pi ft}\, dt$$

$$F(f) = \int_{t_i-\Delta t}^{t_i} A_i \frac{2\pi ft}{(2\pi f)^2 \Delta t_1} e^{-j2\pi ft}\, d(2\pi ft)$$

$$- \int_{t_i-\Delta t_1}^{t_i} A_i \frac{t_i-\Delta t_1}{(2\pi f)\Delta t_1} e^{-j2\pi ft}\, d(2\pi ft)$$

$$+\int_{t_i}^{t_i+\Delta t_1} A_i \frac{t_i+\Delta t_1}{(2\pi f)\Delta t_1} e^{-j2\pi ft}\, d(2\pi ft)$$

$$-\int_{t_i}^{t_i+\Delta t_1} A_i \frac{2\pi ft}{(2\pi f)^2\Delta t_1} e^{-j2\pi ft}\, d(2\pi ft)$$

$$F(f) = \frac{A_i}{(2\pi f)^2\Delta t_1}\left\{e^{-j2\pi ft_i}[j2\pi ft_i+1] - e^{-j2\pi f(t_i-\Delta t_1)}[j2\pi f(t_i-\Delta t_1)+1]\right\}$$

$$- A_i \frac{t_i-\Delta t_1}{(2\pi f)\Delta t_1}\left[je^{-j2\pi ft_i} - je^{-j2\pi f(t_i-\Delta t_1)}\right]$$

$$+ A_i \frac{t_i-\Delta t_1}{(2\pi f)\Delta t_1}\left[je^{-j2\pi f(t_i+\Delta t_1)} - je^{-j2\pi ft_i}\right]$$

$$- \frac{A_i}{(2\pi f)^2\Delta t_1}\left\{e^{-j2\pi f(t_i+\Delta t_1)}[j2\pi f(t_i+\Delta t_1)+1] - e^{-j2\pi ft_i}[j2\pi ft_i+1]\right\}$$

$$F(f) = \frac{A_i}{(2\pi f)^2\Delta t_1}\left[e^{-j2\pi ft_i} - e^{-j2\pi f(t_i+\Delta t_1)} - e^{-j2\pi f(t_i-\Delta t_1)} + e^{-j2\pi ft_i}\right]$$

$$= A_i\,\Delta t_1\, e^{-j2\pi ft_i}\left[\frac{2 - e^{-j2\pi f\Delta t_1} - e^{j2\pi f\Delta t_1}}{(2\pi f\Delta t_1)^2}\right]$$

$$F(f) = A_i\,\Delta t_1\, e^{-j2\pi ft_i}\left(\frac{\sin \pi f\Delta t_1}{\pi f\Delta t_1}\right)^2 .$$

If one were to calculate the spectrum of an isoceles triangel of height A_{i+1}, centered at t_{i+1}, and of width $2\Delta t_1$, it is fairly obvious from the preceding derivation that

$$F(f) = A_{i+1} \Delta t_1 \, e^{-j2\pi f t_{i+1}} \left(\frac{\sin \pi f \Delta t_1}{\pi f \Delta t_1} \right)^2 .$$

Using the summation theorem of Fourier transforms, it is a simple extention of the preceding discussion to show that the spectrum of a straight-line model waveform with n uniform intervals of Δt is

$$F(f) = \left(\frac{\sin \pi f \Delta t}{\pi f \Delta t} \right)^2 \sum_{i=0}^{n} A_i \, \Delta t \, e^{-j2\pi f t_i} .$$

This spectrum is exact for a waveform defined by straight-line segments between uniformly spaced amplitude points, the first and last of which are zero. The spectrum just calculated has application in situations where one might have used a numerical integration, but it gives improved accuracy and removes the high-frequency limitation.

The method applied in the derivation for uniformly spaced intervals can easily be extended to several sets of uniformly spaced intervals. Consider, for example, that we have a waveform approximated by straight lines joining amplitude points spaced Δt_1 apart from t_o to t_k, and by straight lines joining amplitude points spaced Δt_2 apart from t_k to t_n. Assume also that A_o, A_k, and A_n are zero; then it is reasonable to suppose that

$$F(f) = \left(\frac{\sin \pi f \Delta t_1}{\pi f \Delta t_1}\right)^2 \sum_{i=0}^{n} A_i \, \Delta t_1 \, e^{-j2\pi f t_i}$$

$$+ \left(\frac{\sin \pi f \Delta t_2}{\pi f \Delta t_2}\right)^2 \sum_{i=k}^{n} A_i \, \Delta t_2 \, e^{-j2\pi f t_i} \quad .$$

The spectrum is exact for the model waveform as it is defined and can be calculated over all frequencies.

The limitation that the first and last amplitude points of a waveform be zero is no problem, since the waveform must be of finite length, and must therefore be preceded and followed by zero-amplitude level. One is not, however, justified in assuming that the amplitude at an interval-transition point within the waveform will always be zero. When, in fact, the transition amplitude level is not zero, a transition term must be added to the spectrum. At the transition point the triangle is no longer isoceles, as shown in the following sketch, and the spectrum correction is as follows:



D-5585-24

TRANSITION TRIANGLE

$$F(f)_{transition} = \int_{t_k-\Delta t_1}^{t_k} A_k \left[\frac{t-t_k+\Delta t_1}{\Delta t_1}\right] e^{-j2\pi ft}\, dt$$

$$+ \int_{t_k}^{t_k+\Delta t_2} A_k \left[\frac{t_k+\Delta t_2-t}{\Delta t_2}\right] e^{-j2\pi ft}\, dt$$

$$= \int_{t_k-\Delta t_1}^{t_k} A_k \frac{2\pi ft}{(2\pi f)^2 \Delta t_1} e^{-j2\pi ft}\, d(2\pi ft)$$

$$- \int_{t_k-\Delta t_1}^{t_k} A_k \frac{t_k-\Delta t_1}{(2\pi f)\Delta t_1} e^{-j2\pi ft}\, d(2\pi ft)$$

$$+ \int_{t_k}^{t_k+\Delta t_2} A_k \frac{t_k+\Delta t_2}{(2\pi f)\Delta t_2} e^{-j2\pi ft}\, d(2\pi ft)$$

$$- \int_{t_k}^{t_k+\Delta t_2} A_k \frac{2\pi ft}{(2\pi f)^2 \Delta t_2} e^{-j2\pi ft}\, d(2\pi ft)$$

$$F(f)_{transition} = \frac{A_k}{(2\pi f)^2 \Delta t_1} \left\{ e^{-j2\pi ft_k} [j2\pi ft_k + 1] \right.$$

$$\left. - e^{-j2\pi f(t_k-\Delta t_1)} [j2\pi f(t_k-\Delta t_1)+1] \right\}$$

$$- A_k \frac{t_k-\Delta t_1}{(2\pi f)\Delta t_1} \left[je^{-j2\pi ft_k} - je^{-j2\pi f(t_k-\Delta t_1)} \right]$$

$$+ A_k \frac{t_k+\Delta t_2}{(2\pi f)\Delta t_2} \left[je^{-j2\pi f(t_k+\Delta t_2)} - je^{-j2\pi ft_k} \right]$$

$$- \frac{A_k}{(2\pi f)^2 \Delta t_2} \left\{ e^{-j2\pi f(t_k+\Delta t_2)} [j2\pi f(t_k+\Delta t_2)+1] \right.$$

$$\left. - e^{-j2\pi f t_k} [j2\pi f t_k + 1] \right\}$$

$$F(f) = A_k e^{-j2\pi f t_k} \left[\frac{\Delta t_1}{(2\pi f \Delta t_1)^2} \left(1 - e^{j2\pi f \Delta t_1}\right) \right.$$

$$\left. + \frac{\Delta t_2}{(2\pi f \Delta t_2)^2} \left(1 - e^{-j2\pi f \Delta t_2}\right) \right] .$$

It is evident that if $\Delta t_1 = \Delta t_2$, the transition term reduces to the spectrum derived earlier for the isoceles triangle, as it should.

We are now in a position to write the spectrum of any model waveform constructed of straight-line segments joining points in several sets of uniform intervals. Assume, for example, that there are three uniform sample intervals Δt_1, Δt_2, and Δt_3, with transitions at i = k and m. Then

$$F(f) = \left(\frac{\sin \pi f \Delta t_1}{\pi f \Delta t_1} \right)^2 \sum_{i=0}^{k - \frac{\Delta t_1}{\Delta t_1}} A_i \Delta t_1 e^{-j2\pi f t_i}$$

$$+ A_k e^{-j2\pi f t_k} \left[\frac{\Delta t_1}{(2\pi f \Delta t_1)^2} \left(1 - e^{j2\pi f \Delta t_1}\right) + \frac{\Delta t_2}{(2\pi f \Delta t_2)^2} \left(1 - e^{-j2\pi f \Delta t_2}\right) \right]$$

$$+ \left(\frac{\sin \pi f \Delta t_2}{\pi f \Delta t_2} \right)^2 \sum_{i = k + \frac{\Delta t_2}{\Delta t_1}}^{m - \frac{\Delta t_2}{\Delta t_1}} A_i \, \Delta t_2 \, e^{-j2\pi f t_i}$$

$$+ A_m \, e^{-j2\pi f t_m} \left[\frac{\Delta t_2}{(2\pi f \Delta t_2)^2} \left(1 - e^{j2\pi f \Delta t_2} \right) + \frac{\Delta t_3}{(2\pi f \Delta t_3)^2} \left(1 - e^{-j2\pi f \Delta t_3} \right) \right]$$

$$+ \left(\frac{\sin \pi f \Delta t_3}{\pi f \Delta t_3} \right)^2 \sum_{i = m + \frac{\Delta t_3}{\Delta t_1}}^{n} A_i \, \Delta t_3 \, e^{-j2\pi f t_i} \quad .$$

The method can easily be extended to additional sets of uniform scaling intervals and gives an exact answer for the model waveform as it is defined. The preceding equations have been programmed for the CDC 3200 computer at SRI and have provided very satisfactory results.

Many of the concepts developed for the first Fourier integral equation can be applied to the inverse equation. First, however, the equation must be rearranged to avoid the question of negative frequencies.

$$F(t) = \int_{-\infty}^{\infty} F(f) \, e^{j2\pi f t} \, df$$

$$= \int_{-\infty}^{0} F(f) \, e^{j2\pi f t} \, df + \int_{0}^{\infty} F(f) \, e^{j2\pi f t} \, df$$

$$= \int_{0}^{\infty} F(-f) \, e^{-j2\pi f t} \, df + \int_{0}^{\infty} F(f) \, e^{j2\pi f t} \, df \quad .$$

We know that the time functions we are dealing with are real for all values of time; therefore, the spectrum of such a function has the following properties:

$$|F(f)| = |F(-f)|$$

and

$$\arg F(f) = -\arg F(-f) \quad .$$

Furthermore, these properties also tell us that

$$\text{Re } F(f) = \text{Re } F(-f)$$

and

$$\text{Im } F(f) = -\text{Im } F(-f) \quad .$$

It is now possible to separate the second Fourier integral into cartesian coordinate

$$F(t) = \int_0^\infty F(-f)\, e^{-j2\pi ft}\, df + \int_0^\infty F(f)\, e^{j2\pi ft}\, df$$

$$= \int_0^\infty [\text{Re } F(-f) + j \text{ Im } F(-f)]\, [\cos 2\pi ft - j \sin 2\pi ft]\, df$$

$$+ \int_0^\infty [\text{Re } F(f) + j \text{ Im } F(f)]\, [\cos 2\pi ft + j \sin 2\pi ft]\, df$$

$$F(t) = 2 \int_0^\infty [\text{Re } F(f) \cos 2\pi ft - \text{Im } F(f) \sin 2\pi ft]\, df$$

It is possible to make a further simplification, since we know that $F(t)$ should be zero for $t < 0$. That is,

$$F(-t) = 0 = 2 \int_0^\infty \text{Re } F(f) \cos 2\pi f(-t) - \text{Im } F(f) \sin 2\pi f(-t) \quad ,$$

or

$$\int_0^\infty \text{Re } F(f) \cos 2\pi ft \; df = - \int_0^\infty \text{Im } F(f) \sin 2\pi ft \; df \quad .$$

The final form of the second Fourier integral for time functions that are real and are zero for $t < 0$ is

$$F(t) = 4 \int_0^\infty \text{Re } F(f) \cos (2\pi ft) \; df \quad .$$

The two fundamental ideas, straight-line approximation of the input waveform and several sets of uniform intervals, can now be applied to the second Fourier integral in the simplified version just derived. The time function corresponding to the isoceles triangle spectrum shown in the sketch below is as follows:



D-5585-28

$$F(t) = 4 \int_{f_i - \Delta f_1}^{f_i} \text{Re } F(f_i) \left[\frac{f - f_i + \Delta f_1}{\Delta f_1} \right] \cos 2\pi ft \; df$$

$$+ 4 \int_{f_i}^{f_i + \Delta f_1} \text{Re } F(f_i) \left[\frac{f_i + \Delta f_1 - f}{\Delta f_1} \right] \cos 2\pi ft \; df$$

$$F(t) = 4 \int_{f_i-\Delta f_1}^{f_i} \text{Re } F(f_i) \frac{2\pi ft}{(2\pi t)^2 \Delta f_1} \cos 2\pi ft \; d(2\pi ft)$$

$$- 4 \int_{f_i-\Delta f_1}^{f_i} \text{Re } F(f_i) \frac{f_i - \Delta f_1}{(2\pi t)\Delta f_1} \cos 2\pi ft \; d(2\pi ft)$$

$$+ 4 \int_{f_i}^{f_i+\Delta f_1} \text{Re } F(f_i) \frac{f_i + \Delta f_1}{(2\pi t)\Delta f_1} \cos 2\pi ft \; d(2\pi ft)$$

$$- 4 \int_{f_i}^{f_i+\Delta f_1} \text{Re } F(f_i) \frac{2\pi ft}{(2\pi t)^2 \Delta f_1} \cos 2\pi ft \; d(2\pi ft)$$

$$F(t) = 4 \left(\frac{\sin \pi t \Delta f_1}{\pi t \Delta f_1} \right)^2 \text{Re } F(f_i) \; \Delta f_1 \; \cos 2\pi t f_i \quad .$$

From this it is evident that the time function corresponding to a straight-line model of the real part of a spectrum scale at n uniform intervals of Δf_1 would be

$$F(t) = 4 \left(\frac{\sin \pi t \Delta f_1}{\pi t \Delta f_1} \right)^2 \sum_{i=0}^{n} \text{Re } F(f_i) \; \Delta f_1 \; \cos 2\pi f_i \quad .$$

For this to be strictly true requires that Re $F(f_o)$ and Re $F(f_n)$ be zero.

It is now a simple matter to extend the derivation to several sets of uniform intervals. As was the case in the time domain, the transition triangle in the frequency domain is no longer isoceles, as shown in the following sketch:



D-5585-29

TRANSITION TRIANGLE

The transition term is

$$F(t)_{transition} = 4 \int_{f_n - \Delta f_1}^{f_n} \text{Re } F(f_n) \left[\frac{f - f_n + \Delta f_1}{\Delta f_1} \right] \cos 2\pi ft \; df$$

$$+ 4 \int_{f_n}^{f_n + \Delta f_2} \text{Re } F(f_n) \left[\frac{f_n + \Delta f_2 - f}{\Delta f_2} \right] \cos 2\pi ft \; df$$

$$F(t)_{transition} = 4 \int_{f_n - \Delta f_1}^{f_n} \text{Re } F(f_n) \frac{2\pi ft}{(2\pi t)^2 \Delta f_1} \cos 2\pi ft \; d(2\pi ft)$$

$$- 4 \int_{f_n - \Delta f_1}^{f_n} \text{Re } F(f_n) \frac{f_n - \Delta f_1}{(2\pi t) \Delta f_1} \cos 2\pi ft \; d(2\pi ft)$$

$$+ 4 \int_{f_n}^{f_n + \Delta f_2} \text{Re } F(f_n) \frac{f_n + \Delta f_2}{(2\pi t) \Delta f_2} \cos 2\pi ft \; d(2\pi ft)$$

$$- 4 \int_{f_n}^{f_n + \Delta f_2} \text{Re } F(f_n) \frac{2\pi ft}{(2\pi t)^2 \Delta f_2} \cos 2\pi ft \; d(2\pi ft)$$

$$F(t)_{transition} = 4 \text{ Re } F(f_n) \left\{ \frac{\Delta f_1}{(2\pi t \Delta f_1)^2} [\cos 2\pi t f_n - \cos 2\pi t (f_n - \Delta f_1)] \right.$$

$$\left. + \frac{\Delta f_2}{(2\pi t \Delta f_2)^2} [\cos 2\pi t f_n - \cos 2\pi t (f_n + \Delta f_2)] \right\}$$

$$= 4 \text{ Re } F(f_n) \cos 2\pi t f_n \left[\Delta f_1 \frac{1 - \cos 2\pi t \Delta f_1}{(2\pi t \Delta f_1)^2} + \Delta f_2 \frac{1 - \cos 2\pi t \Delta f_2}{(2\pi t \Delta f_2)^2} \right]$$

$$- 4 \text{ Re } F(f_n) \sin 2\pi t f_n \left[\Delta f_1 \frac{\sin 2\pi t \Delta f_1}{(2\pi t \Delta f_1)^2} - \Delta f_2 \frac{\sin 2\pi t \Delta f_2}{(2\pi t \Delta f_2)^2} \right]$$

$$F(t)_{transition} = 4 \text{ Re } F(f_n) \cos 2\pi t_n \left[\frac{\Delta f_1}{2} \left(\frac{\sin \pi t \Delta f_1}{\pi t \Delta f_1} \right)^2 + \frac{\Delta f_2}{2} \left(\frac{\sin \pi t \Delta f_2}{\pi t \Delta f_2} \right)^2 \right]$$

$$- 4 \text{ Re } F(f_n) \sin 2\pi t f_n \left[\Delta f_1 \frac{\sin 2\pi t \Delta f_1}{(2\pi t \Delta f_1)^2} - \Delta f_2 \frac{\sin 2\pi t \Delta f_2}{(2\pi t \Delta f_2)^2} \right] :$$

A special version of the transition term is necessary for the zero frequency term, which generally will be zero.

$$F(t)_{f_o \text{ transition}} = 4 \text{ Re } F(f_o) \frac{\Delta f_1}{2} \left(\frac{\sin \pi t \Delta f_1}{\pi t \Delta f_1} \right)^2 .$$

It is now possible to write an expression for the second Fourier transform, using a straight-line model of the real part of the frequency spectrum scaled with several sets of uniform intervals. For example, for the real part of a spectrum scaled at intervals of Δf_1 from $i=0$ to k, Δf_2 from $i=k$ to m, and Δf_3 from $i=m$ to n, the transform is

$$F(t) = 4 \operatorname{Re} F(f_o) \frac{\Delta f_1}{2} \left(\frac{\sin \pi t \Delta f_1}{\pi t \Delta f_1} \right)^2$$

$$+ 4 \left(\frac{\sin \pi t \Delta f_1}{\pi t \Delta f_1} \right)^2 \sum_{i=1}^{k - \frac{\Delta f_1}{\Delta f_1}} \operatorname{Re} F(f_i) \Delta f_1 \cos 2\pi f_i$$

$$+ 4 \operatorname{Re} F(f_k) \cos 2\pi t f_k \left[\frac{\Delta f_1}{2} \left(\frac{\sin \pi t \Delta f_1}{\pi t \Delta f_1} \right)^2 + \frac{\Delta f_2}{2} \left(\frac{\sin \pi t \Delta f_2}{\pi t \Delta f_2} \right)^2 \right]$$

$$- 4 \operatorname{Re} F(f_k) \sin 2\pi t f_k \left[\Delta f_1 \frac{\sin 2\pi t \Delta f_1}{(2\pi t \Delta f_1)^2} - \Delta f_2 \frac{\sin 2\pi t \Delta f_2}{(2\pi t \Delta f_2)^2} \right]$$

$$+ 4 \left(\frac{\sin \pi t \Delta f_2}{\pi t \Delta f_2} \right)^2 \sum_{i = k + \frac{\Delta f_2}{\Delta f_1}}^{m - \frac{\Delta f_2}{\Delta f_3}} \operatorname{Re} F(f_i) \Delta f_2 \cos 2\pi t_i$$

$$+ 4 \operatorname{Re} F(f_m) \cos 2\pi t f_m \left[\frac{\Delta f_2}{2} \left(\frac{\sin \pi t \Delta f_2}{\pi t \Delta f_2} \right)^2 + \frac{\Delta f_3}{2} \left(\frac{\sin \pi t \Delta f_3}{\pi t \Delta f_3} \right)^2 \right]$$

$$- 4 \operatorname{Re} F(f_m) \sin 2\pi t f_m \left[\Delta f_2 \frac{\sin 2\pi t \Delta f_2}{(2\pi t \Delta f_2)^2} - f_3 \frac{\sin 2\pi t \Delta f_3}{(2\pi t \Delta f_3)^2} \right]$$

$$+ 4 \left(\frac{\sin \pi t \Delta f_3}{\pi t \Delta f_3} \right)^2 \sum_{i = m + \frac{\Delta f_3}{\Delta f_1}}^{n} \operatorname{Re} F(f_i) \Delta t_3 \cos 2\pi t f_i \quad .$$

The above expression for the Fourier transform has been generalized to several sets of uniform scaling intervals and programmed for the CDC 3200 computer at SRI. To date, the results of sample waveforms have been very satisfactory. Some question still remains whether a time function can be represented satisfactorily for a spectrum which is predominately imaginary. If this situation does inhibit accuracy when only the real part of the spectrum is used, then it is planned to revert to the expression

$$F(t) = 2 \int_0^{\infty} [\mathrm{Re}\, F(f) \cos 2\pi ft - \mathrm{Im}\, F(f) \sin 2\pi ft]\, df \quad .$$

The changes required in the equations already programmed for the simplified transform are replacement of the coefficient 4 by the coefficient 2, and replacement of $\mathrm{Re}\, F(f) \cos 2\pi ft$ by

$$\mathrm{Re}\, F(f) \cos 2\pi ft - \mathrm{Im}\, F(f) \sin 2\pi ft \quad .$$

Many of the ideas used in the transform derivations were based on the work of Draper et al.*

---

* Draper, C. S., W. McKay, and S. Lee, Instrument Engineering, Vol. II (McGraw-Hill Book Co., Inc., New York, 1953).

APPENDIX B

```
      PROGRAM FORPLEX(INPUT,OUTPUT)
      COMPLEX RCVC
      COMMON/CBFOR/OM(1500),RCVC(1500),T(500),V(500),VN(500)
      DIMENSION DT(10),TIM(10)
      SIG=.1
      NT=1
      DT(1)=.02
      TIM(1)=2.5
      JFREQ=0
      CALL FIPI(SIG,NT,DT,TIM,JFREQ)
      JGO=1
      PRINT 500
500   FORMAT(*  DELTA T,ENDING TIME PT*)
      PRINT 505,(DT(I),TIM(I),I=1,NT)
505   FORMAT(2E15.4)
300   DO 2004 I=1,JFREQ
      RCVC(I)=(1.,0.)/CMPLX(1.,OM(I))
2004  CONTINUE
      GO TO (200,201)JGO
200   CALL FIOFI(JFREQ,SIG)
      JGO=2
      GO TO 300
201   CALL FICO(SUM,JTIM)
      PRINT 501,SUM,SIG
501   FORMAT(1H1,* RMS DIFFERENCE=*E12.4,5X,*CRIT=*E12.4)
      PRINT 503
503   FORMAT(//10X*TIME*6X*PREV CALC*11X*CALC*11X*DIFF*6X*COMP. FN.*)
      DO 202 I=1,JTIM
      VDIF=VN(I)-V(I)
      CMPFN=EXP(-T(I))
      PRINT 502,T(I),VN(I),V(I),VDIF,CMPFN
502   FORMAT(5E15.5)
202   CONTINUE
      STOP
      END
```

```
      SUBROUTINE FIPI(SIG,NT,DT,TIM,JFREQ)
      COMPLEX RCVC
      COMMON/CBFOR/OM(1500),RCVC(1500),T(500),V(500),VN(500)
      COMMON/RNGE/WRANGE(31),DELTAW(31),DW(31),EW(31),J2
      COMMON/RMT/MT
      COMMON/TIME/TDD(10),XKK(10),TD(20),XK(20)
      DIMENSION DT(NT),TIM(NT)
      DO 4 I=1,NT
      TDD(I)=DT(I)
4     XKK(I)=TIM(I)
6075  NN=NT+1
      DO 2002 L=2,NN
      TD(L)=TDD(L-1)
2002  XK(L)=XKK(L-1)
      XK(1)=0.
      T(1)=0.
      K2=1
      DO 6003 I=2,NN
      K1=K2+1
      ZK=(XK(I)-XK(I-1))/TD(I)+K2
      K2=ZK+.1
      DO 6003 K=K1,K2
6003  T(K)=T(K-1)+TD(I)
      J2=K2
      TMIN=1.E30
      DO 1 I=1,NT
1     IF(TMIN.GT.TDD(I))TMIN=TDD(I)
      IWU=ALOG10(1./SIG/TMIN)+1.
      WU=10.**IWU
      IWL=ALOG10(SIG/XKK(NT))-1.
      DWL=10.**IWL
      WL=DWL/SIG
      IWL=ALOG10(WL)
      MT=(IWU-IWL+1)*3
      CALL WRSET(WL,SIG)
      CALL WSET(J)
      JFREQ=J
      RETURN
      END
```

```
      SUBROUTINE WRSET(WL,SIG)
      COMPLEX RCVC
      COMMON/CBFOR/OM(1500),RCVC(1500),T(500),V(500),VN(500)
      COMMON/RNGE/WRANGE(31),DELTAW(31),DW(31),EW(31),J2
      COMMON/RMT/MT
      DW(1)=0.
      EW(1)=0.
      WRANGE(1)=WL/5.
      WRANGE(2)=WL/2.
      WRANGE(3)=WL
      DELTAW(1)=SIG*WL/10.
      DELTAW(2)=2.*DELTAW(1)
      DELTAW(3)=5.*DELTAW(1)
      DO 1 I=4,MT,3
      WRANGE(I+2)=10.*WRANGE(I-1)
      DELTAW(I)=SIG*WRANGE(I-1)
      WRANGE(I)=2.*WRANGE(I-1)
      DELTAW(I+1)=2.*DELTAW(I)
      WRANGE(I+1)=5.*WRANGE(I-1)
1     DELTAW(I+2)=5.*DELTAW(I)
      DO 2 I=1,MT
      DW(I+1)=DELTAW(I)
2     EW(I+1)=WRANGE(I)
      PRINT 4
4     FORMAT(*1   DELTA W, W RANGE*)
      PRINT 3,(DELTAW(I),WRANGE(I),I=1,MT)
3     FORMAT(2E20.6)
      RETURN
      END
```

```
      SUBROUTINE WSET(J)
      COMPLEX RCVC
      COMMON/CBFOR/OM(1500),RCVC(1500),T(500),V(500),VN(500)
      COMMON/RNGE/WRANGE(31),DELTAW(31),DW(31),EW(31),J2
      COMMON/RMT/MT
      COMMON/CX/TSRC(31),TRC(31),KX(31),TSR(31),TSI(31),
     1TR(31),TI(31),JX(31),TSX(31),TSX2(31),XX(31),FD(31)
      M=MT+1
      K2=1
      OM(1)=0.
      DO 1052 I=2,M
      K1=K2+1
      K2=K2+(EW(I)-EW(I-1))/DW(I)+.1
      DO 1051 J=K1,K2
1051  OM(J)=OM(J-1)+DW(I)
1052  JX(I-1)=K1-1
      MM=K2
      JX(MT+1)=K2
      J=MM
      PRINT 11,J,MT
11    FORMAT(* NO. OMEGA PTS.=*I6,5X*NO. OMEGA PRS. =*I5)
      RETURN
      END
```

```
      SUBROUTINE FIOFI(JFREQ,SIG)
      COMPLEX RCVC
      COMMON/CBFOR/OM(1500),RCVC(1500),T(500),V(500),VN(500)
      COMMON/RNGE/WRANGE(31),DELTAW(31),DW(31),EW(31),J2
      COMMON/RMT/MT
      CALL CALCLT
      DO 1 I=1,J2
1     VN(I)=V(I)
      WL=WRANGE(3)/10.
      MT=MT+6
      SIG=SIG/2.
      CALL WRSET(WL,SIG)
      CALL WSET(J)
      JFREQ=J
      RETURN
      END
```

```
      SUBROUTINE FICO(SUM,JTIM)
      COMPLEX RCVC
      COMMON/CBFOR/OM(1500),RCVC(1500),T(500),V(500),VN(500)
      COMMON/RNGE/WRANGE(31),DELTAW(31),DW(31),EW(31),J2
      COMMON/RMT/MT
      CALL CALCLT
      SUMV=0.
      SUMVNV=0.
      DO 1 I=2,J2
      RSUM=VN(I)-V(I)
      VSUM=V(I)
      DT=T(I)-T(I-1)
      SUMV=SUMV+VSUM*VSUM*DT
    1 SUMVNV=SUMVNV+RSUM*RSUM*DT
      SUM=SQRT(SUMVNV/SUMV)
      JTIM=J2
      RETURN
      END
```

```
      SUBROUTINE CALCLT
      COMPLEX RCVC
      COMMON/CBFOR/OM(1500),RCVC(1500),T(500),V(500),VN(500)
      COMMON/RNGE/WRANGE(31),DELTAW(31),DW(31),EW(31),J2
      COMMON/RMT/MT
      COMMON/TIME/TDD(10),XKK(10),TD(20),XK(20)
      COMMON/CX/TSRC(31),TRC(31),KX(31),TSR(31),TSI(31),
     1TR(31),TI(31),JX(31),TSX(31),TSX2(31),XX(31),FD(31)
      S1=.9999999999
      S2=-.16666666609
      S3=8.333330731E-3
      S4=-1.9840833822E-4
      S5=2.752401177E-6
      S6=-2.386893E-8
      PI=3.14159265358979
      TPI=2.*PI
      RADDG=180./PI
      DEGRD=PI/180.
      DO 2003 I=1,MT
      DW(I)=DELTAW(I)
 2003 EW(I+1)=WRANGE(I)
10352 DO 500 K=1,J2
      RVP=0.
      TRC(MT)=0.
      DO 1045 L=1,MT
      FD(L)=DW(L)/TPI
      X=TPI*FD(L)*T(K)/2.
      XX(L)=X
      IF(ABS(X)-PI/2.)1042,1043,1044
 1042 TSX(L)=(S1+X*X*(S2+X*X*(S3+X*X*(S4+X*X*(S5+X*X*S6)))))
      GO TO 1045
 1043 TSX(L)=2./PI
      GO TO 1045
 1044 TSX(L)=SIN(X)/X
 1045 TSX2(L)=TSX(L)*TSX(L)
      TS1C=RCVC(1)*FD(1)*TSX2(1)
      EW(1)=0.
      JX(1)=1
      DO 1039 L=1,MT
      TS4C=0.
      JX(L+1)=JX(L)+(EW(L+1)-EW(L))/DW(L)+.1
      K1=JX(L)+1
      K2=JX(L+1)
      DO 600 J=K1,K2
      IF(J.EQ.K2) GO TO 1046
 600  TS4C=TS4C+FD(L)*(REAL(RCVC(J))*COS(OM(J)*T(K))
     1-AIMAG(RCVC(J))*SIN(OM(J)*T(K)))
 1046 TSRC(L)=2.*TSX2(L)*TS4C
      IF(L.EQ.MT) GO TO 1039
 1057 IF(K.NE.1) GO TO 1056
 1058 TRC(L)=(REAL(RCVC(J))*COS(OM(J)*T(K))-AIMAG(RCVC(J))*SIN(OM(J)*T
     1(K)))*(FD(L)*TSX2(L)+FD(L+1)*TSX2(L+1))
```

```
      GO TO 1039
1056  TRC(L)=2.*((REAL(RCVC(J))*COS(OM(J)*T(K))-AIMAG(RCVC(J))*SIN
     1(OM(J)*T(K)))*(.5*FD(L)*TSX2(L)+.5*FD(L+1)*TSX2(L+1))
     2-REAL(RCVC(J))*SIN(OM(J)*T(K))*(FD(L)*TSX(L)/XX(L)-FD(L+1)
     3*TSX(L+1)/XX(L+1)))
1039  RVP=RVP+TSRC(L)+TRC(L)
      V(K)=RVP+TS1C
  500 CONTINUE
      RETURN
      END
```

DELTA W, W RANGE

1.000000E-03	2.000000E-02
2.000000E-03	5.000000E-02
5.000000E-03	1.000000E-01
1.000000E-02	2.000000E-01
2.000000E-02	5.000000E-01
5.000000E-02	1.000000E+00
1.000000E-01	2.000000E+00
2.000000E-01	5.000000E+00
5.000000E-01	1.000000E+01
1.000000E+00	2.000000E+01
2.000000E+00	5.000000E+01
5.000000E+00	1.000000E+02
1.000000E+01	2.000000E+02
2.000000E+01	5.000000E+02
5.000000E+01	1.000000E+03

NO. OMEGA PTS. = 186 NO. OMEGA PRS. = 15
DELTA T, ENDING TIME PT
2.0000E-02 2.5000E+00

DELTA W, W RANGE

5.000000E-05	2.000000E-03
1.000000E-04	5.000000E-03
2.500000E-04	1.000000E-02
5.000000E-04	2.000000E-02
1.000000E-03	5.000000E-02
2.500000E-03	1.000000E-01
5.000000E-03	2.000000E-01
1.000000E-02	5.000000E-01
2.500000E-02	1.000000E+00
5.000000E-02	2.000000E+00
1.000000E-01	5.000000E+00
2.500000E-01	1.000000E+01
5.000000E-01	2.000000E+01
1.000000E+00	5.000000E+01
2.500000E+00	1.000000E+02
5.000000E+00	2.000000E+02
1.000000E+01	5.000000E+02
2.500000E+01	1.000000E+03
5.000000E+01	2.000000E+03
1.000000E+02	5.000000E+03
2.500000E+02	1.000000E+04

NO. OMEGA PTS. = 511   NO. OMEGA PRS. = 21

RMS DIFFERENCE = 1.0489E-02 CRIT = 5.0000E-02

TIME	PREV CALC	CALC	DIFF	COMP. FN.
0.	5.00096E-01	5.00074E-01	2.23040E-05	1.00000E+00
2.00000E-02	9.72465E-01	9.81389E-01	-8.92335E-03	9.80199E-01
4.0000E-02	9.58092E-01	9.63008E-01	-4.91588E-03	9.60789E-01
6.00000E-02	9.37527E-01	9.43080E-01	-5.55237E-03	9.41765E-01
8.00000E-02	9.36038E-01	9.27096E-01	8.94138E-03	9.23116E-01
1.00000E-01	9.02164E-01	9.00069E-01	2.09458E-03	9.04837E-01
1.20000E-01	8.80579E-01	8.88089E-01	-7.50967E-03	8.86920E-01
1.40000E-01	8.73926E-01	8.68947E-01	4.97963E-03	8.69358E-01
1.60000E-01	8.54081E-01	8.51475E-01	2.60540E-03	8.52144E-01
1.80000E-01	8.37244E-01	8.36741E-01	5.02500E-04	8.35270E-01
2.00000E-01	8.25208E-01	8.19581E-01	5.62657E-03	8.18731E-01
2.20000E-01	8.05453E-01	8.03136E-01	2.31705E-03	8.02519E-01
2.40000E-01	7.82676E-01	7.86126E-01	-3.45025E-03	7.86628E-01
2.60000E-01	7.62224E-01	7.68987E-01	-6.76320E-03	7.71052E-01
2.80000E-01	7.59814E-01	7.54982E-01	4.83234E-03	7.55784E-01
3.00000E-01	7.44019E-01	7.44674E-01	-6.55598E-04	7.40818E-01
3.20000E-01	7.25737E-01	7.23621E-01	2.11601E-03	7.26149E-01
3.40000E-01	7.18009E-01	7.13274E-01	4.73501E-03	7.11770E-01
3.60000E-01	6.93002E-01	6.99390E-01	-6.38775E-03	6.97676E-01
3.80000E-01	6.73588E-01	6.76939E-01	-3.35095E-03	6.83861E-01
4.00000E-01	6.72610E-01	6.72364E-01	2.45656E-04	6.70320E-01
4.20000E-01	6.65077E-01	6.61725E-01	3.35253E-03	6.57047E-01
4.40000E-01	6.51748E-01	6.42197E-01	9.55068E-03	6.44036E-01
4.60000E-01	6.40395E-01	6.36005E-01	4.39006E-03	6.31284E-01
4.80000E-01	6.21387E-01	6.21190E-01	1.96887E-04	6.18783E-01
5.00000E-01	5.99928E-01	6.00541E-01	-6.12963E-04	6.06531E-01
5.20000E-01	5.91465E-01	5.94789E-01	-3.32335E-03	5.94521E-01
5.40000E-01	5.86963E-01	5.86242E-01	7.20434E-04	5.82748E-01
5.60000E-01	5.75895E-01	5.70121E-01	5.77340E-03	5.71209E-01
5.80000E-01	5.62076E-01	5.61258E-01	8.18056E-04	5.59898E-01
6.00000E-01	5.43260E-01	5.50021E-01	-6.76112E-03	5.48812E-01
6.20000E-01	5.25368E-01	5.31548E-01	-6.17998E-03	5.37944E-01
6.40000E-01	5.15335E-01	5.21915E-01	-6.58039E-03	5.27292E-01
6.60000E-01	5.11742E-01	5.19110E-01	-7.36742E-03	5.16851E-01
6.80000E-01	5.09952E-01	5.06824E-01	3.12831E-03	5.06617E-01
7.00000E-01	5.02443E-01	4.97195E-01	5.24805E-03	4.96585E-01
7.20000E-01	4.87649E-01	4.90774E-01	-3.12557E-03	4.86752E-01
7.40000E-01	4.75388E-01	4.74320E-01	1.06823E-03	4.77114E-01
7.60000E-01	4.67147E-01	4.63848E-01	3.29867E-03	4.67666E-01
7.80000E-01	4.61682E-01	4.61760E-01	-7.74881E-05	4.58406E-01

TIME	PREV CALC	CALC	DIFF	COMP. FN.
8.00000E-01	4.60433E-01	4.53242E-01	7.19078E-03	4.49329E-01
8.20000E-01	4.53654E-01	4.43802E-01	9.85286E-03	4.40432E-01
8.40000E-01	4.38278E-01	4.35994E-01	2.28385E-03	4.31711E-01
8.60000E-01	4.22966E-01	4.22447E-01	5.19514E-04	4.23162E-01
8.80000E-01	4.11322E-01	4.09729E-01	1.59283E-03	4.14783E-01
9.00000E-01	4.02815E-01	4.05030E-01	-2.21494E-03	4.06570E-01
9.20000E-01	3.98390E-01	4.00545E-01	-2.15581E-03	3.98519E-01
9.40000E-01	3.93679E-01	3.92123E-01	1.55603E-03	3.90628E-01
9.60000E-01	3.83377E-01	3.84478E-01	-1.09582E-03	3.82893E-01
9.80000E-01	3.72234E-01	3.74930E-01	-2.69543E-03	3.75311E-01
1.00000E+00	3.64486E-01	3.63014E-01	1.47209E-03	3.67879E-01
1.02000E+00	3.59313E-01	3.57950E-01	1.36353E-03	3.60595E-01
1.04000E+00	3.57351E-01	3.56333E-01	1.01774E-03	3.53455E-01
1.06000E+00	3.54398E-01	3.50786E-01	3.61212E-03	3.46456E-01
1.08000E+00	3.47372E-01	3.44136E-01	3.23639E-03	3.39596E-01
1.10000E+00	3.37729E-01	3.35119E-01	2.60958E-03	3.32871E-01
1.12000E+00	3.27410E-01	3.23903E-01	3.50720E-03	3.26280E-01
1.14000E+00	3.18560E-01	3.16453E-01	2.10689E-03	3.19819E-01
1.16000E+00	3.11014E-01	3.13038E-01	-2.02373E-03	3.13486E-01
1.18000E+00	3.03307E-01	3.08805E-01	-5.49811E-03	3.07279E-01
1.20000E+00	2.94347E-01	3.02204E-01	-7.85659E-03	3.01194E-01
1.22000E+00	2.84745E-01	2.93043E-01	-8.29860E-03	2.95230E-01
1.24000E+00	2.76031E-01	2.83023E-01	-6.99185E-03	2.89384E-01
1.26000E+00	2.69473E-01	2.75920E-01	-6.44686E-03	2.83654E-01
1.28000E+00	2.65338E-01	2.72783E-01	-7.44543E-03	2.78037E-01
1.30000E+00	2.62722E-01	2.71611E-01	-8.88839E-03	2.72532E-01
1.32000E+00	2.60482E-01	2.69018E-01	-8.53526E-03	2.67135E-01
1.34000E+00	2.58327E-01	2.62656E-01	-4.32927E-03	2.61846E-01
1.36000E+00	2.56075E-01	2.55025E-01	1.04968E-03	2.56661E-01
1.38000E+00	2.53711E-01	2.49477E-01	4.23357E-03	2.51579E-01
1.40000E+00	2.51224E-01	2.46174E-01	5.05014E-03	2.46597E-01
1.42000E+00	2.47695E-01	2.44775E-01	2.91962E-03	2.41714E-01
1.44000E+00	2.43080E-01	2.43236E-01	-1.55732E-04	2.36928E-01
1.46000E+00	2.38150E-01	2.36926E-01	1.22389E-03	2.32236E-01
1.48000E+00	2.32495E-01	2.27879E-01	4.61569E-03	2.27638E-01
1.50000E+00	2.26398E-01	2.21075E-01	5.32273E-03	2.23130E-01
1.52000E+00	2.20492E-01	2.16116E-01	4.37650E-03	2.18712E-01
1.54000E+00	2.14832E-01	2.13180E-01	1.65205E-03	2.14381E-01
1.56000E+00	2.09527E-01	2.12529E-01	-3.00240E-03	2.10136E-01
1.58000E+00	2.05775E-01	2.08535E-01	-2.76012E-03	2.05975E-01
1.60000E+00	2.03793E-01	2.01165E-01	2.62713E-03	2.01897E-01
1.62000E+00	2.01555E-01	1.96117E-01	5.43769E-03	1.97899E-01
1.64000E+00	1.99619E-01	1.92729E-01	6.89042E-03	1.93980E-01

TIME	PREV CALC	CALC	DIFF	COMP. FN.
1.66000E+00	1.97799E-01	1.91015E-01	6.78344E-03	1.90139E-01
1.68000E+00	1.94743E-01	1.91635E-01	3.10888E-03	1.86374E-01
1.70000E+00	1.91739E-01	1.89655E-01	2.08327E-03	1.82684E-01
1.72000E+00	1.88967E-01	1.83570E-01	5.39637E-03	1.79066E-01
1.74000E+00	1.84815E-01	1.77288E-01	7.52741E-03	1.75520E-01
1.76000E+00	1.79327E-01	1.72381E-01	6.94567E-03	1.72045E-01
1.78000E+00	1.73134E-01	1.68612E-01	4.52213E-03	1.68638E-01
1.80000E+00	1.66233E-01	1.66536E-01	-3.03858E-04	1.65299E-01
1.82000E+00	1.59621E-01	1.63957E-01	-4.33666E-03	1.62026E-01
1.84000E+00	1.54362E-01	1.58097E-01	-3.73532E-03	1.58817E-01
1.86000E+00	1.50041E-01	1.51507E-01	-1.46594E-03	1.55673E-01
1.88000E+00	1.46442E-01	1.47077E-01	-6.34732E-04	1.52590E-01
1.90000E+00	1.43525E-01	1.44456E-01	-9.31112E-04	1.49569E-01
1.92000E+00	1.41350E-01	1.44025E-01	-2.67537E-03	1.46607E-01
1.94000E+00	1.40413E-01	1.44374E-01	-3.96086E-03	1.43704E-01
1.96000E+00	1.40331E-01	1.42399E-01	-2.06851E-03	1.40858E-01
1.98000E+00	1.40563E-01	1.39010E-01	1.55253E-03	1.38069E-01
2.00000E+00	1.40913E-01	1.36100E-01	4.81337E-03	1.35335E-01
2.02000E+00	1.40225E-01	1.34396E-01	5.82899E-03	1.32655E-01
2.04000E+00	1.38222E-01	1.33740E-01	4.48207E-03	1.30029E-01
2.06000E+00	1.36153E-01	1.32954E-01	3.19895E-03	1.27454E-01
2.08000E+00	1.33457E-01	1.30497E-01	2.95992E-03	1.24930E-01
2.10000E+00	1.30114E-01	1.25779E-01	4.33526E-03	1.22456E-01
2.12000E+00	1.26967E-01	1.20728E-01	6.23931E-03	1.20032E-01
2.14000E+00	1.23319E-01	1.17004E-01	6.31445E-03	1.17655E-01
2.16000E+00	1.19153E-01	1.14456E-01	4.69762E-03	1.15325E-01
2.18000E+00	1.15564E-01	1.12929E-01	2.63525E-03	1.13042E-01
2.20000E+00	1.12758E-01	1.11166E-01	1.59112E-03	1.10803E-01
2.22000E+00	1.10360E-01	1.08215E-01	2.14563E-03	1.08609E-01
2.24000E+00	1.08637E-01	1.05302E-01	3.33465E-03	1.06459E-01
2.26000E+00	1.07097E-01	1.03373E-01	3.72418E-03	1.04350E-01
2.28000E+00	1.04891E-01	1.02517E-01	2.37387E-03	1.02284E-01
2.30000E+00	1.02460E-01	1.02446E-01	1.43794E-05	1.00259E-01
2.32000E+00	9.98930E-02	1.01713E-01	-1.82043E-03	9.82736E-02
2.34000E+00	9.70114E-02	9.96046E-02	-2.59319E-03	9.63276E-02
2.36000E+00	9.39910E-02	9.64330E-02	-2.44201E-03	9.44202E-02
2.38000E+00	9.04027E-02	9.29622E-02	-2.55955E-03	9.25506E-02
2.40000E+00	8.61149E-02	8.99145E-02	-3.79963E-03	9.07180E-02
2.42000E+00	8.14341E-02	8.72313E-02	-5.79726E-03	8.89216E-02
2.44000E+00	7.69447E-02	8.44447E-02	-7.49996E-03	8.71609E-02
2.46000E+00	7.31982E-02	8.12072E-02	-8.00899E-03	8.54350E-02
2.48000E+00	7.03922E-02	7.77855E-02	-7.39331E-03	8.37432E-02
2.50000E+00	6.83020E-02	7.48621E-02	-6.56006E-03	8.20850E-02